Exercise 4.4#11

First, the expressions for the piecewise definitions.

```
> fa:=x^3+1;
```

$$fa := x^3 + 1$$

```
> fb:=1.001+0.03*(x-0.1)+0.3*(x-0.1)^2+2*(x-0.1)^3;
```

$$fb := .998 + .03\,x + .3\,(x - .1)^2 + 2\,(x - .1)^3$$

```
> fc:=1.009+0.15*(x-0.2)+0.9*(x-0.2)^2+2*(x-0.2)^3;
```

$$fc := .979 + .15\,x + .9\,(x - .2)^2 + 2\,(x - .2)^3$$

Verification of continuity.

```
> subs(x=0.1,[fa,fb]);
```

$$[1.001, 1.001]$$

```
> subs(x=0.2,[fb,fc]);
```

$$[1.009, 1.009]$$

Equality means that the expression is continuous. Now for the calculation of derivatives.

```
> fa1:=diff(fa,x);fb1:=diff(fb,x);fc1:=diff(fc,x);
```

$$fa1 := 3\,x^2$$

$$fb1 := -.03 + .6\,x + 6\,(x - .1)^2$$

$$fc1 := -.21 + 1.8\,x + 6\,(x - .2)^2$$

Continutity of derivatives.

```
> subs(x=0.1,[fa1,fb1]);subs(x=0.2,[fb1,fc1]);
```

$$[.03, .03]$$

$$[.15, .15]$$

Again, results are equal, so first derivatives are contiuous. Now look at second derivatives.

```
> fa2:=diff(fa1,x);fb2:=diff(fb1,x);fc2:=diff(fc1,x);
```

$$fa2 := 6\,x$$

$$fb2 := -.6 + 12\,x$$

$$fc2 := -.6 + 12\,x$$

```
> subs(x=0.1,[fa2,fb2]);subs(x=0.2,[fb2,fc2]);
```

$$[.6, .6]$$

$$[1.8, 1.8]$$

Continuity verified. That means that this expression is a cubic spline. To see what kind of spline it is, look at the boundary conditions.

```
> subs(x=0,[fa,fa1,fa2]);
```

$$[1, 0, 0]$$

```
> subs(x=0.3,[fc,fc1,fc2]);
```

$$[1.035, .39, 3.0]$$

The derivatives are zero at the left endpoint, but the values at the right endpoint have no particular pattern. It is reasonable to assume that this a clamped spline in which the vlaues of the first derivatives at the endpoints were based on known prooerties of the data. To get a better idea of the appearance of this function, look at a graph. The equality of the expressions for fb2 and fc2 suggests that fb and fc are actually the same cubic. This is the so-called "not a knot" condition used to deal with missing boundary conditions. See the reference books mentioned in the textbook for more details. We can verify the equality of fb and fc.

```
> expand(fb-fc);
```

$$0$$

Verified!

```
> plot(piecewise(x<0.1,fa,x<0.2,fb,fc),x=0..0.3);
```



Note that the bottom of the graph is at 1, and the horizontal scale is about 10 time the vertical scale, so this curve is much tamer than this picture suggests. The average value of the function seems to be about 1.01, so the integral will be about 0.303. We can calculate the integral exactly by integrating separately over each piece and summing the results.

```
> Actual:=int(fa,x=0..0.1)+int(fb,x=0.1..0.2)+int(fc,x=0.2..0.3);
```

$$Actual := .3024250000$$

Six point trapezoidal rule:

```
> h:=0.3/6;
```

$$h := .05000000000$$

```
> Trap:=(h/2)*(subs(x=0,fa)+2*subs(x=0.05,fa)+2*subs(x=0.1,fa)+2*subs(x=0.15,fb)+2*subs(x=0.2,fc)+2*subs(x=0.25,fc)+subs(x=0.3,fc));
```

$$Trap := .3025062500$$

To estimate the error, we need to know something about the second derivative of this piecewise function. Each piece of the second derivative was an increasing linear function, so the second derivative is largest at 0.3, where it is given by fc2.

```
> E:=(0.3/12)*h^2*subs(x=0.3,fc2);
```

$$E := .0001875000000$$

```
> Actual-Trap;
```

-.0000812500

```
> %/E;
```

-.4333333333

The ratio is lees than 1 in absolute value,since we expected to be overestimating the size of the error. It is negative since the graph is concave upward throughot the interval, so the trapezoidal rule must be larger than the integral.

Six point Simpson's rule:

```
> (h/3)*(subs(x=0,fa)+4*subs(x=0.05,fa)+2*subs(x=0.1,fa)+4*subs(x=0.
  15,fb)+2*subs(x=0.2,fc)+4*subs(x=0.25,fc)+subs(x=0.3,fc));
```

.3024250000

This value is more than just a good approximation to the integral of the piecewise cubic; it is the exact integral. This follows from th fact that Simpson's rule gives an exact answer for any cubic polynomial. Although we have collected terms, the expression could be written as a sum of applications of Simpson's rule on each pices of the domain where the function is given by one of the cubic polynomials.